
Resumo – Cidades com elevado grau de impermeabilização são geralmente susceptíveis a 

extravasamentos em seu sistema de drenagem quando da ocorrência de chuvas intensas. Uma 

possível medida para evitar esse tipo de problema é a construção de reservatórios de detenção e 

retenção. A análise e dimensionamento dessas estruturas hidráulicas requer a solução da equação 

diferencial ordinária que representa a conservação de massa, em conjunto com o hidrograma que 

representa as vazões afluentes à bacia de detenção e com as equações de orifícios e vertedores, que 

são as estruturas normalmente empregadas para a saída da bacia. O presente trabalho propõe um 

código livre para a solução da referida equação, desenvolvido em linguagem reconhecida pelo 

Scilab, que também é um software livre. Como parte do desenvolvimento do código, foi elaborada 

uma equação empírica para o cálculo do coeficiente de descarga em orifícios retangulares. O código 

foi verificado com a realização de testes numéricos empregando um programa semelhante e com 

dados relativos a uma bacia de detenção concebida para a cidade de São Carlos-SP, tendo 

apresentado desvios menores que 1%. Uma segunda comparação foi realizada com uma 

metodologia adimensional disponível na literatura, tendo sido encontrados resultados com desvios 

inferiores a 5%. 

 

Palavras-Chave – bacia de detenção, drenagem urbana, hidráulica. 

 

INTRODUÇÃO 

Eventos hidrológicos capazes de produzir chuvas intensas em cidades com elevada 

impermeabilização têm como consequência a ocorrência de extravasamentos dos canais de 

macrodrenagem. As inundações urbanas, por sua vez, podem acarretar transtornos relacionados à 

saúde pública, afogamentos e danos aos bens materiais públicos e privados. Eventos como esses 

foram relatados em Santana (2021), que em seu trabalho expõe a situação do rio Jaguaribe em 

Salvador - BA e apresenta possíveis soluções com o uso de bacias de detenção. As bacias de 

detenção e retenção estão entre as possíveis soluções capazes de atenuar e retardar os picos dos 

hidrogramas das ondas de cheia. Bacias de retenção são definidas como aquelas que preservam 

certo volume de água em seu interior; enquanto que as bacias de detenção permanecem vazias 

quando cessa o escoamento através de suas estruturas hidráulicas (PORTO, 2002). 

A análise e o dimensionamento dessas bacias envolvem o uso da equação de conservação de 

massa em conjunto com equações capazes de modelar orifícios e vertedores, que são as estruturas 

empregadas para proporcionar a saída da água da bacia de detenção. A vazão afluente é dada pelo 

hidrograma afluente, sendo aproximado numericamente por meio de interpolações ou, quando 

possível, através de funções, como por exemplo, a distribuição normal descrita pela equação de 

Gauss (PORTO, 2002). A formulação governante resultante da modelagem matemática é uma 

equação diferencial ordinária cuja solução analítica geral é desconhecida, considerando a variedade 

de hidrogramas afluentes possíveis. 

Os métodos numéricos utilizados para a integração da equação diferencial incluem os 

métodos clássicos do Cálculo Numérico, como o método de Euler e os métodos de Runge-Kutta. Na 

literatura de Hidrologia, como Chow et al. (1988) e Tucci (2005), cita-se também o método de Puls 

(1928). Porto (2002) adimensionalizou a equação diferencial para as condições de bacias de 

detenção simples e duais, bacias com saída composta por orifícios e vertedores, tendo obtido 

soluções numéricas para diferentes condições de projeto, possibilitando o cálculo do volume da 

bacia a partir da vazão máxima efluente pretendida após a sua construção. 

Simões et al. (2015) desenvolveram um código em Matlab
®
 com o uso do método de Runge e 

Kutta de quarta ordem capaz de simular o trânsito da onda de cheia na bacia de detenção. Simões et 

al. (2017) elaboraram um código com a linguagem do mesmo software comercial e com o método 

de Runge-Kutta de quinta ordem de Butcher (1964, 2003, p. 92). Os códigos propostos por esses 

autores simulam o trânsito de ondas de cheia em bacias de detenção a partir de informações 

geométricas da bacia. Desse modo, podem servir aos propósitos de análise de uma bacia pré-

existente ou ao dimensionamento de uma nova bacia por tentativa e erro.  

Embora os códigos citados sejam de acesso gratuito, eles necessitam do programa Matlab
®
, 

cuja licença gratuita está presentemente limitada ao uso por um período de trinta dias. O objetivo do 

presente trabalho foi desenvolver um código para o cálculo do trânsito da onda de cheia na bacia de 

detenção empregando o método numérico de Runge-Kutta de quarta ordem, no ambiente do 

software livre Scilab 6.0.1, que é um software gratuito com características semelhantes às do 

Matlab
®
. Como objetivo adicional propôs-se o desenvolvimento de uma formulação para o cálculo 

do coeficiente de descarga de orifícios retangulares. 

 

MATERIAIS E MÉTODOS 

Modelo matemático para o problema 

 

A lei física básica para a modelagem do problema de dimensionamento da bacia de detenção é 

a equação de conservação de massa. A mesma é aplicada a um reservatório de nível variável, com a 

cota da superfície livre considerada uniforme. A equação de conservação de massa em sua forma 

euleriana e integral deve ser escrita para um volume de controle com superfície de controle 

deformável, com a seguinte forma: 

 

0 =
𝑑

𝑑𝑡
∭ 𝜌𝑑𝑉𝑜𝑙

𝑉𝐶
+∬ 𝜌𝑣𝑟⃗⃗  ⃗. �⃗� 𝑑𝐴𝑆𝐶

,      (1) 

 

em que t = tempo, VC = volume de controle,  = massa específica da água, Vol = volume, SC = 

superfície de controle, 𝑣𝑟⃗⃗  ⃗ = 𝑣 − 𝑣𝐴⃗⃗⃗⃗  é a velocidade relativa do fluido em relação à superfície de 

controle, 𝑣  é a velocidade absoluta do fluido, 𝑣𝐴⃗⃗⃗⃗  é a velocidade da superfície de controle, �⃗�  = vetor 

unitário normal à superfície de controle, A = área da superfície de controle. 

 

Admite-se a incompressibilidade para o escoamento na bacia de detenção, condição que 

implica  = constante, portanto, 
 

0 =
𝑑

𝑑𝑡
∭ 𝑑𝑉𝑜𝑙

𝑉𝐶
+∬ 𝑣𝑟⃗⃗  ⃗. �⃗� 𝑑𝐴𝑆𝐶

.      (2) 

O volume de controle é definido com uma forma tal que a sua parte superior coincida com a 

superfície livre da água As superfícies laterais e de fundo devem coincidir com as paredes do 

reservatório e com as superfícies das estruturas hidráulicas de entrada e saída na bacia de detenção. 

Desse modo, apenas a superfície de controle superior movimenta-se acompanhando a superfície da 

água, resultando em uma velocidade relativa igual a zero, isto é, v = vA. Sendo assim, a partir da 

equação 2, escreve-se: 

 

0 =
𝑑𝑉𝑜𝑙

𝑑𝑡
−∬ 𝑣𝑑𝐴

𝑒𝑛𝑡𝑟𝑎𝑑𝑎𝑠
+∬ 𝑣𝑑𝐴

𝑠𝑎í𝑑𝑎𝑠
.     (3) 

 

Sendo a vazão igual à integral de vdA, escreve-se: 

 
𝑑𝑉𝑜𝑙

𝑑𝑡
= 𝑄𝑒 − 𝑄𝑠.        (4) 

 

Na equação 4, Qe é igual a soma das vazões de entrada e Qs é igual a soma das vazões de 

saída. As vazões afluentes são representadas matematicamente pelo hidrograma afluente à bacia de 

detenção. Os orifícios e vertedores empregados como estruturas de descarga e controle fornecem as 

equações para a vazão efluente resultante, Qs. O volume, Vol, pode ser modelado analiticamente. 

No caso de uma bacia prismática ele se resume simplesmente a Vol = hAR, sendo h a altura de 

escoamento no reservatório e AR a área da superfície livre. É possível também estabelecer uma 

função Vol = f(h) que, no código numérico, pode ser aproximada por interpolações lineares. 

 

A equação 5, empregada para modelar o escoamento através do orifício, e a equação 6, 

escrita para o vertedor retangular de parede fina, são aquelas resultantes do uso da equação de 

Bernoulli e conservação de massa, seguido de correções para a contração do jato e para a dissipação 

de energia, com a introdução dos coeficientes de contração, Cc e de velocidade, Cv, cujo produto é 

igual ao coeficiente de descarga, Cd, conforme descrito em Porto (2006). Nos primeiros instantes, o 

orifício funciona com h < a (ver definição na Figura 1). Essa condição foi aproximada como se o 

mesmo fosse um vertedor retangular. Para h ≥ a, a teoria dos grandes orifícios é utilizada. O 

vertedor retangular de parede delgada inicia seu funcionamento se h ≥ P, como descrito pela 

equação 6. 

 

{
𝑄𝑜 =

2

3
𝐶𝑑𝑜√2𝑔𝑏ℎ

3/2, 𝑠𝑒 ℎ < 𝑎,

𝑄𝑜 =
2

3
𝐶𝑑𝑜√2𝑔𝑏[ℎ

3/2 − (ℎ − 𝑎)
3

2], 𝑠𝑒 ℎ ≥ 𝑎,
    (5) 

 

{𝑄𝑣 =
2

3
𝐶𝑑𝑣√2𝑔𝐿𝑣ℎ𝑣

3

2 , 𝑠𝑒 ℎ ≥ 𝑃,

𝑄𝑣 = 0, 𝑠𝑒 ℎ < 𝑃.
 Sendo hv=h - P   (6) 

 
Figura 1 – Parede de uma bacia de detenção dual, com orifício e vertedor retangular de parede delgada. 

 

L

a

P

hb

Métodos numéricos 

 

Empregando série de Taylor, Carl Runge (1856-1927) e Martin W. Kutta (1867-1944) 

desenvolveram métodos de integração numérica de 1ª, 2ª, 3ª e 4ª ordens. Esses métodos 

transformam a equação diferencial original, no presente caso a equação 4, em um sistema de 

equações algébricas. O método de Runge e Kutta de quarta ordem foi usado neste estudo, sendo a 

estrutura do algoritmo proposto apresentada a seguir: 

 

𝑆𝑒𝑗𝑎 𝐹(𝑡, ℎ(𝑡)) =  𝑄𝑒 − 𝑄𝑠(ℎ(𝑉𝑜𝑙(𝑡))), calcule: 

{
 
 
 
 

 
 
 
 

𝑘1 = 𝐹(𝑡𝑖 , 𝑉𝑜𝑙𝑖),

𝐼𝑛𝑡𝑒𝑟𝑝𝑜𝑙𝑒: ℎ (𝑉𝑜𝑙𝑖 + 𝑘1
Δ𝑡

2
) ,

𝑘2 = 𝐹 (𝑡𝑖 +
Δ𝑡

2
, 𝑉𝑜𝑙𝑖 + 𝑘1

Δ𝑡

2
) ,

𝐼𝑛𝑡𝑒𝑟𝑝𝑜𝑙𝑒: ℎ (𝑉𝑜𝑙𝑖 + 𝑘2
Δ𝑡

2
) ,

𝑘3 = 𝐹 (𝑡𝑖 +
Δ𝑡

2
, 𝑉𝑜𝑙𝑖 + 𝑘2

Δ𝑡

2
) ,

𝐼𝑛𝑡𝑒𝑟𝑝𝑜𝑙𝑒: ℎ(𝑉𝑜𝑙𝑖 + 𝑘3Δ𝑡)

𝑘4 = 𝐹(𝑡𝑖 + Δ𝑡, 𝑉𝑜𝑙𝑖 + 𝑘3Δ𝑡).

      (7) 

 

Com os valores de k1, k2, k3 e k4, calcule Voli+1 empregando a equação 8: 

 

𝑉𝑜𝑙𝑖+1 = 𝑉𝑜𝑙𝑖 +
Δ𝑡

6
(𝑘1 + 2𝑘2 + 2𝑘3 + 𝑘4)     (8) 

 
Em seguida, calcule hi+1 com uma função de interpolação aplicada à curva cota-volume do 

reservatório. O código desenvolvido faz uso de uma função de interpolação linear disponível na 

biblioteca do Scilab. 

 

Equação para cálculo do coeficiente de descarga em orifícios retangulares 

 

Em seu livro, Bos (1989, p.273) afirma que, para orifícios retangulares, submersos e com 

contração completa, o coeficiente de descarga é Cd = 0,61. Segundo Lencastre (1996) e Porto 

(2006), para a condição de descarga livre, Cd depende da forma, dimensões e da carga hidráulica do 

orifício. Os dados de Fanning apresentados por King (1939), citado por Lencastre (1996), são 

apresentados na Tabela 1 e indicam a dependência de Cd para com h e a altura do orifício. 

 
Tabela 1 – Coeficientes de descarga para orifícios retangulares e verticais. 

 
Fonte: adaptado de Fonte: dados de Fanning, citado por King (1939), citado por Lencastre (1996, p.598). 

 

h Altura do orifício em [mm], com largura de 0,30 m e com contração completa

[m] 38 75 150 225 300 450 600 1200

0,12 0,625 0,619 - - - - - -

0,15 0,624 0,618 0,615 - - - - -

0,30 0,622 0,616 0,611 0,608 0,605 0,608 - -

0,60 0,619 0,614 0,609 0,606 0,604 0,606 0,609 -

0,90 0,616 0,612 0,608 0,605 0,603 0,605 0,607 0,609

1,20 0,614 0,610 0,607 0,604 0,603 0,604 0,606 0,608

1,50 0,612 0,609 0,605 0,603 0,602 0,604 0,605 0,606

3,00 0,606 0,604 0,602 0,601 0,601 0,601 0,602 0,603

6,00 0,607 0,604 0,602 0,601 0,601 0,601 0,602 0,603

15,00 0,614 0,607 0,605 0,604 0,603 0,603 0,606 0,609

Observando as formas das distribuições dos pontos formados por Cd e h/a, Meira (2020) propôs a 

equação 9: 

 

𝐶𝑑 = 𝑐1 [(ℎ/𝑎)
𝑐2 +

𝑐3

(1+ℎ/𝑎)𝑐4
]       (9) 

 

Meira (2020) calculou os valores de c1, c2, c3 e c4 minimizando a soma dos erros quadráticos 

entre os valores calculados com a equação 9 e os dados experimentais utilizando a ferramenta 

Solver do MS Excel
®
. A Figura 2 possibilita a comparação dos resultados obtidos com a equação 9 

e os dados experimentais. Para valores de a/b intermediários, foi implementada no código uma 

rotina para interpolação linear, com base na função disponível no Scilab. Na Tabela 2 são 

apresentados os coeficientes de correlação, R, calculados considerando os dados experimentais e os 

valores de Cd calculados com a equação 9. Todos os coeficientes de correlação resultaram maiores 

que 0,97, indicando a adequação da equação proposta para o problema em questão. Os valores 

calculados e os dados experimentais possuem aderência entre eles, com erro relativo máximo de 

0,526% entre os valores calculados e os dados de Fanning, valor 5,5 vezes menor em relação ao 

erro relativo calculado com o valor médio dos dados, que é 0,608. 

 
Tabela 2 – Coeficientes c1 a c4 em função de a/b e os coeficientes de correlação, R, entre os valores calculados de Cd e 

os dados de Fanning. 

 

Fonte: Meira (2020). 

 

Figura 2 – Comparação entre os dados experimentais e os valores calculados, para Cd, com a equação 9. 

 
Fonte: Meira (2020). 

 

RESULTADOS E DISCUSSÃO 

 

Código 

 

O código proposto, apresentado no Quadro 1, foi escrito no software livre Scilab, em sua 

versão 6.0.1. A estrutura do código pode ser entendida como a composição de: (1) pré-

processamento; (2) processamento e (3) pós-processamento. Na parte de pré-processamento, o 

a/b 0,127 0,25 0,5 0,75 1 1,5 2 4

c1 1,64736E-06 3,43742E-06 1,08741E-04 2,45323E-05 1,40000E-05 1,72759E-05 2,94601E-05 4,83100E-04

c2 1,531 1,481 1,144 1,427 1,548 1,747 1,860 1,450

c3 384722,18 181148,69 5651,02 24915,05 43320,47 35286,71 20776,79 1269,36

c4 0,0095268 0,0074179 0,0182090 0,0066846 0,0039766 0,0067811 0,0090236 0,0146630

R 0,981 0,988 0,970 0,992 0,982 0,981 0,972 0,972

0,60

0,61

0,62

0,63

0,60 0,61 0,62 0,63

C
d

(C
al

cu
la

d
o

)

Cd (experimental)

Ajuste exato

usuário deve digitar o hidrograma afluente em duas matrizes, uma para as vazões e outra para os 

tempos correspondentes às vazões. Em seguida, deve digitar, também em duas matrizes, a curva 

cota-volume. Para o código apresentado no Quadro 1, foi inserida uma curva cota-volume definida 

por uma lei de potência correspondente ao exemplo proposto por Porto (2002), referente à parte alta 

da bacia do córrego do Gregório, em São Carlos – SP. Esse exemplo também foi explorado por 

Simões et al. (2015) utilizando o software comercial Matlab
®

, e foi adotado no presente trabalho 

para realização de comparações. Os demais dados encontram-se no pré-processamento apresentado 

no Quadro 1. Ademais, comentários foram inseridos em verde (com a sintaxe //comentário). 

 

 
Quadro 1 – Código para análise e dimensionamento de bacias de detenção. 

clc 

clear 

close 

//PRÉ-PROCESSAMENTO 

// Vazões do hidrograma de entrada: 

Qe1 = [0; 0.07; 0.51; 1.53; 2.56; 4.90; 8.35; 11.42; 14.87; 16.92; 17.72; 20.43; 23.43; 25.17; 24.67; 23.13; 19.46; 13.15; 10.21; 

8.23; 6.17; 4.19; 3.01; 1.61]; 

 

//Instantes correspondentes às vazões Qe: 

t = [0.00; 341.23; 494.79; 648.34; 750.71; 921.33; 1160.19; 1381.99; 1689.10; 1876.78; 1996.21; 2269.19; 2627.49; 3240.00; 

3565.88; 3890.05; 4316.59; 4896.68; 5220.85; 5459.72; 5800.95; 6193.38; 6483.42; 7182.96]; 

 

//Curva cota-volume: 

hVol = 0:1:6; 

Volh = 5000.*hVol.^1.25; 

 

[a1,a2] = size(t); 

tt = t(a1,1);//Tempo total, em [s] 

Nt = 5000; //Número de nós do eixo temporal 

dt = tt/(Nt-1);//intervalo temporal 

g=9.8;//aceleração devido à gravidade [m/s²] 

 

//Para cálculos com R-K de 4ª ordem: 

vol = zeros(Nt,1);//vetor para armazenar os volumes 

h = zeros(Nt,1); //Vetor para armazenar as profundidades (bacia de detenção) 

h0 = h(1,1); //valor inicial de h 

Qo = zeros(Nt,1);//vazão através do orifício 

Qv = zeros(Nt,1);//vazão através do vertedor 

 

scf(1)//preserva o gráfico durante após execução do código 

plot(t/60, Qe1, 'o') //Hidrograma original 

xlabel('t [min]') 

ylabel('Q [m^3/s]') 

legend('Hidrograma original'); 

 

tint=linspace(0,tt,Nt); //É gerada uma malha com número de nós igual a Nt e de 0 a tt, tint. 

Qe1int=interp1(t,Qe1(:,1),tint); 

Qe1int=Qe1int';//Cálculo da transposta. 

 

scf(2) 

plot(hVol, Volh, 'o') 

xlabel('h [m]') 

ylabel('Vol [m^3]') 

legend('Curva cota-volume'); 

 

//Nova interpolação. Observe que o método requer a realização de cálculos em t+(1/2)dt, sendo esta uma possível estratégia para 

programação. 

dt12=dt/2;//Metade do intervalo temporal 
Nt12=tt/dt12+1;//Número de nós do eixo temporal correspondente a dt12 

tint12=linspace(0,tt,Nt12);//Gera matriz com números linearmente espaçados conforme Nt12 

Qe1int12=interp1(tint,Qe1int(:,1),tint12);//Interpolação linear das vazões do hidrograma 

Qe1int12=Qe1int12';//Cálculo da transposta para compatibilidade com a matriz temporal 

 

//Gráficos: 

scf(3) 

plot(t/60, Qe1, 'o', tint/60, Qe1int) 

xlabel('t [min]') 

ylabel('Q [m^3/s]') 

legend('Hidrograma original', 'Hidrograma interpolado'); 

//Estruturas Hidráulicas - saídas: 

//Orifício retangular (teoria dos grandes orifícios): 

a = 1; 

b = 1.508; 

Ao = a*b;//Área 

 

//Para o cálculo de Cdo: 

//Valores tabelados: 

tab_ab = [0.127; 0.25; 0.50; 0.75; 1.0; 1.5; 2; 4];  

tab_c1 = [0.00000164736; 0.00000343742; 0.000108741;0.0000245323;0.000014; 0.0000172759; 0.0000294601; 0.0004831]; 

tab_c2 = [1.531; 1.481; 1.144;1.427;1.548;1.747;1.86;1.45]; 

tab_c3 = [384722.18;181148.69;5651.02;24915.05;43320.47;35286.71;20776.79;1269.36]; 

tab_c4 = [0.0095268;0.0074179;0.018209;0.0066846;0.0039766;0.0067811;0.0090236;0.014663]; 

 

//Cálculo dos coeficientes a serem utilizados: 

c1=interp1(tab_ab, tab_c1, a/b); 

c2=interp1(tab_ab, tab_c2, a/b); 

c3=interp1(tab_ab, tab_c3, a/b); 

c4=interp1(tab_ab, tab_c4, a/b); 

 

//Vertedor de parede delgada: 

Lv = 2;//Largura da soleira 

 

Cdv = 0.72;//coeficiente de vazão adimensional 

P = 3.3;//Altura do vertedor, em [m] 

 

//PROCESSAMENTO: 

k=2;//Contador 

t12=0; 

 

for t=0:dt:(tt-dt) 

//Decisão sobre os cálculos de Qo e Qv com base nas restrições correspondentes: 

        if h(k-1,1)>a then 

            do1=1; 

            do2=0; 

        else 

            do1=0; 

            do2=1; 

        end 

        if h(k-1,1)>P then 

            dv = 1; 

        else 

            dv = 0; 

        end 

 

//Cálculo dos índices do elemento t12: 

t12=t+(1/2)*dt; 

teste12=abs(tint12-t12); 

[aa,n12]=min(teste12);//armazena em n12 a posição de t12, necessário para k2 e k3 

 

k1 = Qe1int(k-1,1)-Qo(k-1,1)-Qv(k-1,1);//k1: método de Runge-Kutta 

 

volki=real(vol(k-1,1)+(1/2)*k1*dt);//uso da função real como exigência da sintaxe do programa. 

 

hh=interp1(Volh, hVol, volki);//Interpolação linear do valor de h correspondente a volki 

 

ho=do1*(hh^1.5 - (hh-a)^1.5) + do2*hh^1.5;//parte do sistema de equações do orifício 

hv=dv*(hh-P);// Carga sobre a soleira do vertedor 

 

Cdo = (c1*(((hh/a)^c2)+(c3/((1+hh/a)^c4))));//Coeficiente de descarga 

 

//k2: usa a vazão interpolada, em t12, e as equações do orifício e vertedor: 

 

k2 = Qe1int12(n12,1)-((2/3)*Cdo*((2*g)^0.5)*b*ho+(2/3)*Cdv*((2*g)^0.5)*Lv*hv^(3/2)); 

 
hh=interp1(Volh, hVol, real(vol(k-1,1)+(1/2)*k2*dt));//Interpolação linear de h correspondente ao volume em t12 calculado com 

k2 

 

ho=do1*(hh^1.5 - (hh-a)^1.5) + do2*hh^1.5; 

hv=dv*(hh-P); 

 

Cdo = (c1*(((hh/a)^c2)+(c3/((1+hh/a)^c4)))); 

 

k3 = Qe1int12(n12,1)-((2/3)*Cdo*((2*g)^0.5)*b*ho+(2/3)*Cdv*((2*g)^0.5)*Lv*hv^(3/2)); 

 

hh=interp1(Volh, hVol, real(vol(k-1,1)+k3*dt));//Interpolação linear de h para o vol correspondente a k3 

ho=do1*(hh^1.5 - (hh-a)^1.5) + do2*hh^1.5; 

hv=dv*(hh-P); 

 

Cdo = (c1*(((hh/a)^c2)+(c3/((1+hh/a)^c4)))); 

 

k4 = Qe1int(k,1)-((2/3)*Cdo*((2*g)^0.5)*b*ho+(2/3)*Cdv*((2*g)^0.5)*Lv*hv^(3/2)); 

 

//Atualização (R-K 4ª ordem): 

vol(k,1)=vol(k-1,1)+(dt/6)*(k1+2*k2+2*k3+k4); 

 

//Interpolação para cálculo de h(k,1): 

h(k,1)=interp1(Volh, hVol, real(vol(k,1))); 

 

ho=do1*((h(k,1))^1.5 - ((h(k,1))-a)^1.5) + do2*(h(k,1))^1.5;  

hv=dv*(h(k,1)-P); 

 

Cdo = (c1*((((h(k,1))/a)^c2)+(c3/((1+(h(k,1)/a))^c4)))); 

 

Qo(k,1) = (2/3)*Cdo*((2*g)^0.5)*b*ho; 

Qv(k,1) = dv*(2/3)*Cdv*((2*g)^0.5)*Lv*hv^(3/2); 

 

k=k+1 

 

end 

 

//Conferindo o valor máximo de Cdo 

[teste1,teste2] = max(h) 

Cdo = (c1*(((teste1/a)^c2)+(c3/((teste1/a)^c4))))  

 

 

//PÓS-PROCESSAMENTO: 

scf(4) 

plot(tint/60,h) 

xlabel('t [min]') 

ylabel('h [m]') 

 

//Integração numérica para cálculo do volume: 

Qt = Qo+Qv; 

[aa,bb]=max(real(Qt)); 

Volume1 = inttrap(tint(1,1:bb),Qe1int(1:bb,1))-inttrap(tint(1,1:bb),Qt(1:bb,1));//Trapézios 

 

//1) Hidrogramas: 

scf(5) 

plot(tint'/60, Qe1int, tint'/60, Qt) 

xlabel('t [min]') 

ylabel('Q [m^3/s]') 

legend('Hidrograma afluente', 'Hidrograma efluente'); 

 

//2) Volume=f(t): 

scf(6) 

plot(tint'/60, vol(:,1)) 

xlabel('t [min]') 

ylabel('Volume [m^3]') 

 

 

Considerando um coeficiente de descarga constante e igual a 0,60 para o orifício, os 

resultados obtidos para o exemplo proposto estão apresentados na Figura 2. Simões et al. (2015) 

empregaram uma interpolação cúbica hermitiana por partes, tendo para isto utilizado  uma função 

disponível no Matlab
®
. No presente trabalho, foi adotada uma interpolação linear, uma vez que o 

Scilab não dispõe do método de interpolação de Hermite, cujos resultados mostraram-se muito 

bons. 

 

A Figura 2a mostra o hidrograma afluente original sobreposto ao interpolado. Observa-se 

que a interpolação linear representa adequadamente o hidrograma original, ou seja, sem a inserção 

de comportamentos espúrios, como pode ocorrer com o uso de interpolação por spline. No mesmo 

gráfico, é apresentado também o hidrograma efluente calculado com o método de Runge-Kutta de 

quarta ordem. Simões et al. (2015) calcularam a vazão máxima efluente igual a 18,7 m³/s, em t = 

73,2 min. Empregando o código proposto neste trabalho, a vazão máxima efluente resultou igual a 

18,6 m³/s, para o mesmo tempo  t = 73,2 min, o que corresponde a um desvio de apenas 0,54% para 

a vazão em relação ao uso da interpolação de Hermite. A comparação entre os valores de h, 

apresentada na Figura 2b, mostra haver excelente aderência entre os resultados calculados com os 

dois códigos e métodos de interpolação, com max(h) = 5,1 m (SIMÕES et al., 2015) e max(h) = 

5,08 m (presente trabalho), com um desvio de 0,39% em relação ao uso da interpolação hermitiana. 

 

(a) (b) 

Figura 2 – Solução numérica obtida com o código proposto e comparações com o código de Simões et al. (2015): (a) 

vazões; (b) altura de escoamento no reservatório. 

 

A equação 9, proposta para o cálculo do coeficiente de descarga do orifício, resultou em um 

valor máximo de h indicado como max(h) = 4,53 m, correspondendo a um volume máximo 

indicado como  max(Vol) = 33.126 m³. Para este problema Porto (2002) calculou max(h) = 4,75 m 

e max(Vol) = 34.633 m³. Comparando os resultados, tendo os de Porto (2002) como referência, os 

desvios relativos correspondem a 4,6% para max(h) e 4,4% para max(Vol), resultado considerado 

apropriado para este dispositivo. 

 

 

CONCLUSÕES 

 

O preente estudo fornece, para uso livre por parte do leitor interessado, o código para 

dimensionamento de bacia de detenção dual e não prismática, uma solução cuja aplicação tem se 

tornado mais viável com o crescimento da impermeabilização das áreas urbanas e o consequente 

crescimento dos picos de cheias.  

Adicionalmente, o código permite que seja utilizada uma equação para obtenção do 

coeficiente de descarga de orifícios retangulares, a qual reproduz muito bem os valores tabelados 

encontrados na literatura. 

Além do código em si, são mostrados os elementos básicos que o compõem, desde a 

equação diferencial governante, as equações de orifícios e vertedores, o método de Runge Kutta 

para a sua solução, permitindo ao interessado entender os desenvolvimentos que permitem o 

dimensionamento desejado.  

Considerando o aumento dos casos de inundação das áreas urbanas de municípios 

espalhados pelo país e a necessidade de soluções a serem implementadas pelas autoridades, 

entende-se que o acesso livre a códigos que permitam o estudo de soluções é urgente. O presente 

estudo insere-se no contexto do presente momento e disponibiliza o código de uma dessas soluções.  

